# $Id$
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=gmetadom
pkgver=0.2.6
pkgrel=6
pkgdesc="C++ bindings for GDome2 DOM implementation"
arch=('i686' 'x86_64')
url="http://gmetadom.sourceforge.net/"
license=('LGPL')
depends=('gdome2')
makedepends=('pkgconfig' 'ocaml')
options=('!makeflags')
source=("http://downloads.sourceforge.net/sourceforge/${pkgname}/${pkgname}-${pkgver}.tar.gz"
        'gmetadom-0.2.6-gcc43.patch')
md5sums=('7bfb5d9c6f20ffa5e790b63d1f8cb481'
         'ea0271007f8650733029327823353d0d')

build() {
  cd ${srcdir}/${pkgname}-${pkgver}
  patch -Np1 -i ${srcdir}/gmetadom-0.2.6-gcc43.patch
  export LDFLAGS=${LDFLAGS/,--as-needed}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make DESTDIR=${pkgdir}/ install
}
